SPICE Models for Vacuum-Tube Amplifiers*

W. MARSHALL LEACH, JR., *AES Fellow*

Georgia Institute of Technology, School of Electrical and Computer Engineering, Atlanta, GA 30332-0250, USA

PSpice analog behavioral models for the vacuum-tube triode and pentode are given. The application of these models to the computer analysis of audio power amplifiers is described. An example SPICE simulation of a vacuum-tube amplifier is presented.

0 INTRODUCTION

In his paper lamenting the demise of the Heathkit [1], Avery Comarow wrote: "The soft glow of a vacuum tube will soon be relegated to nostalgia and the memories of codgers like me." Like Mr. Comarow, the demise of the vacuum tube in audio electronics has been predicted by engineers for years. However, the vacuum tube has enjoyed its niche in the audio field despite the rapid advances in solid-state devices. In particular, the vacuum-tube amplifier has always dominated the msucial instrument amplifier business, where it is an accepted fact that "guitarists prefer tube amps" [2]. In addition, the vacuum-tube amplifier currently enjoys a renaissance in the home hi-fidelity market.

Computer-aided design tools such as SPICE are used almost universally in the design of solid-state electronic circuits. Indeed, many integrated-circuit designers rely exclusively on SPICE simulations to eliminate the need for breadboarding circuits before they are fabricated. The ability to do this with vacuum-tube circuits can be a tremendous boon to designers, provided the device models for vaccum tubes are readily available. This paper describes mathematical models for the vacuum-tube triode and pentode devices, which are suitable for use in computer simulation software. The analog behavioral modeling feature of *PSpice* is used to implement the models in SPICE [3]. (*PSpice* for personal computers is a software product of the MicroSim Corporation.) Tutorial material is included to support the work. Several example simulations are presented to illustrate the use of the models in calculating the frequency response, transient response, and harmonic distortion of amplifier circuits.

The notation used for voltages and currents in this paper corresponds to the following conventions: Total instantaneous values are indicated by a lowercase letter with an uppercase subscript, such as i_P and v_{GK}. DC bias values are indicated by an uppercase letter with an uppercase subscript, such as I_P and V_{GK}. Small-signal ac values are indicated by a lowercase letter with a lowercase subscript, such as i_p and v_{gk}.

1 THE TRIODE

The triode is a vacuum tube that has three active elements. Fig. 1(a) shows the circuit symbol. For simplicity, the heater or filament is not shown. The active elements are the plate P (also called the anode), the grid G, and the cathode K. For the device to be operated as an amplifier, the plate-to-cathode voltage must be positive and the grid-to-cathode voltage is usually nega-



Fig. 1. (a) Circuit symbol for triode. (b) Small-signal model of triode. (c) *PSpice* model for triode.

* Manuscript received 1994 August 20; revised 1994 November 1.

tive. When these conditions are met, the total instantaneous plate current i_P is approximately given by [4]

$$i_P = K(\mu v_{GK} + v_{PK})^{3/2}, \qquad \text{for } \mu v_{GK} + v_{PK} \geq 0$$
$$= 0, \qquad \text{for } \mu v_{GK} + v_{PK} < 0 \quad (1)$$

where K and μ are constants determined by the particular tube. The constant μ is called the amplification factor. It represents the negative ratio of a change in plate voltage to a change in grid voltage under the condition that the plate current remains constant. When v_{GK} is negative, the grid current can usually be neglected. If v_{GK} is positive, grid current flows.

The small-signal plate current is given by

$$i_p = \frac{\partial I_P}{\partial V_{GK}} v_{gk} + \frac{\partial I_P}{\partial V_{PK}} v_{pk}$$
$$= g_m v_{gk} + \frac{v_{pk}}{r_p} \quad (2)$$

where the transconductance g_m and the plate resistance r_p are given by

$$g_m = \frac{\partial I_P}{\partial V_{GK}} = \frac{3\mu I_P}{2(\mu V_{GK} + V_{PK})} \quad (3)$$

$$r_p = \left(\frac{\partial I_P}{\partial V_{PK}}\right)^{-1} = \frac{\mu}{g_m}. \quad (4)$$

The small-signal model of the triode is given in Fig. 1(b). The capacitors in the circuit model the interelectrode capacitances. If the grid is biased at a voltage that is positive with respect to the cathode, the grid current cannot be neglected. In this case a resistor must be added from the grid to the cathode to model the small-signal grid-to-cathode resistance.

Fig. 1(c) shows a *PSpice* subcircuit model for the triode. The model is valid only for $v_{PK} \geq 0$. A series solid-state diode D_1 and resistor R_G approximately model the grid circuit for $v_{GK} > 0$. The SPICE code for the model is given in the following. Parameters which must be assigned numerical values are enclosed in quotations. The PWR and PWRS functions are *PSpice* functions that are used to implement the nonlinear equation for the plate current i_P. The PWR function is defined by PWR(x, m) = $|x|^m$. Similarly, the PWRS function is defined by PWRS(x, m) = $|x|^m$ for $x \geq 0$ and PWRS(x, m) = $-|x|^m$ for $x < 0$. With these definitions, it follows that (PWR(x, m) + PWRS(x, m))/2 = x^m for $x \geq 0$ and (PWR(x, m) + PWRS(x, m))/2 = 0 for $x < 0$.

```
*TRIODE SUBCIRCUIT
.SUBCKT TRIODE P G K
E1 2 0 VALUE = {V(P,K) + "μc"*V(G,K)}
R1 2 0 1K
GP P K VALUE = {"K"*(PWR)(V(2),1.5)
   + PWRS(V(2),1.5))/2}
RGK G 1 "RGK"
D1 1 K DM
CGK G K "Cgk"
CGP G P "Cgp"
CPK P K "Cpk"
.MODEL DM D
.ENDS
```

Fig. 2(a) shows the plot of the plate current versus plate voltage for a typical 12AX7 triode calculated by SPICE with the subcircuit. The curves are shown for a grid-to-cathode voltage v_{GK} that varies from 0 to -5 V in steps of -0.5 V. The 12AX7 parameters used in the subcircuit have the values $K = 1.73 \times 10^{-6}$ and $\mu = 83.5$. Fig. 2(b) shows the plot of the plate current versus plate voltage for a typical 12AT7 triode. The curves are shown for a grid-to-cathode voltage v_{GK} that varies from 0 to -10 V in steps of -1 V. The 12AT7 parameters used in the subcircuit have the values $K = 8.31 \times 10^{-6}$ and $\mu = 44.8$.

The model parameters used for the 12AX7 and the 12AT7 were derived from curves of tube characteristics given in [5]. For the 12AX7, K and μ were determined so that the calculated curves matched those given in [5] at the points ($V_{GK} = 0$, $V_{PK} = 150$ V, $I_P = 3.2$ mA) and ($V_{GK} = -5$ V, $V_{PK} = 450$ V, $I_P = 0.32$ mA). For the 12AT7, K and μ were determined so that the calculated curves matched those given in [5] at the points ($V_{GK} = 0$, $V_{PK} = 200$ V, $I_P = 24$ mA) and ($V_{GK} = -10$ V, $V_{PK} = 550$ V, $I_P = 3.1$ mA). The calculated curves show excellent agreement with those given in [5].

Fig. 3 shows the circuit diagram of a single-stage resistance-coupled triode amplifier, which is used here to illustrate an application of the triode model. The element values are taken from a table of values in [5] for resistance-coupled amplifiers using the 3AV6, 4AV6, 6AV6, 6EU6, 12AV6, 12AX7A, 20EX7, and 7025 "high-mu" vacuum tubes. The circuit is specified to have an inverting voltage gain of 59 (35.4 dB) and a peak output voltage of 34 V. The SPICE code for the analysis is given in the following. Disabled lines are preceded by an asterisk.

```
SINGLE-STAGE TRIODE AMPLIFIER
VPP 4 0 DC 180V
VIN 1 0 AC 1V
*.PARAM X = 1
*VIN 1 0 SIN(0,{-1*X},1K,0,0,0)
RK 2 0 3.5K
RP 3 4 220K
RG 5 0 470K
CK 1 0 2.1U
C 3 5 0.006U
XT1 3 1 2 12AX7
.OP
.AC DEC 20 10 10MEG
*.TRAN (0.01M,2M,0,0.01M)
*.STEP LIN PARAM X 0.2 1 0.2
*.FOUR 1K V(5)
.SUBCKT 12AX7 P G K
```

```
E1 2 0 VALUE = {V(P,K) + 83.5*V(G,K)}
R1 2 0 1K
GP P K VALUE = {1.73E-6*(PWR(V(2),1.5)
   + PWRS(V(2),1.5))/2}
CGP G P 1.7P
CGK G K 1.6P
CPK P K 0.4P
.ENDS
.PROBE
.END.
```

Fig. 4 shows the small-signal ac gain versus frequency calculated by SPICE for the circuit. The midband gain is 35.1 dB, which agrees with the gain specified in [5] within 0.3 dB. The −3-dB cutoff frequencies are 75 Hz and 1.6 MHz. Fig. 5 shows the transient response for a 1-kHz sine wave applied to the amplifier input. A family of curves is displayed which were obtained with the parametric analysis feature of *PSpice*. The curves correspond to peak input signal voltages of 0.2, 0.4, 0.6, 0.8, and 1.0 V. The two horizontal lines are drawn at ±34 V, that is, at the specified maximum peak output voltage for the circuit. The figure shows that the positive signal peaks are hard clipped for the 1.0-V input voltage case. The clipped portion of the waveform is tilted because the amplifier output is ac coupled.

The waveforms exhibit a larger negative peak signal swing than positive peak signal swing. This indicates the presence of strong even-order harmonic distortion. The effect especially becomes obvious when the negative peak swing exceeds the horizontal line at −34 V. As part of the transient analysis, a SPICE distortion analysis (called a Fourier analysis) was used to calculate the distortion components in the waveform. Table 1 summarizes the percent second harmonic distortion and the percent total harmonic distortion predicted by this analysis. It can be seen that the second harmonic distortion is almost equal to the total harmonic distortion for each case. This indicates that the principal distortion created by the amplifier is second order. It has been postulated by many that this could be the principal reason for the difference in the sound of vacuum-tube and solid-state amplifiers.

2 THE PENTODE

The pentode is a vacuum tube which has five active elements. Fig. 6(a) shows the circuit symbol. For simplicity, the heater or filament is not shown. The active elements are the plate P (also called the anode), the control grid G, the screen grid S, the suppressor grid, and the cathode K. The device is normally operated with the suppressor grid connected to the cathode, as shown in the figure. For it to be operated as an amplifier, the plate-to-cathode voltage and the screen-to-cathode voltage must be positive and the grid-to-cathode voltage is usually negative. When these conditions are met and the plate-to-cathode voltage is not too low, the total instantaneous space current i_1 is approximately given by [4],

$$
\begin{aligned}
i_1 &= K(\mu_c v_{GK} + \mu_s v_{SK} + v_{PK})^{3/2}\,, && \text{for } \mu_c v_{GK} + \mu_s v_{SK} + v_{PK} \geq 0 \\
&= 0, && \text{for } \mu_c v_{GK} + \mu_s v_{SK} + v_{PK} < 0
\end{aligned}
\qquad (5)
$$



(a)



(b)

Fig. 2. Simulated triode plate characteristics. (a) 12AX7 triode. (b) 12AT7 triode.

where K is a constant, μ_s is the screen grid amplification factor, and μ_c is the control grid amplification factor.

The plate current i_P and the screen current i_S are given by

$$i_P = ai_1 \tag{6}$$

$$i_S = (1 - a)\, i_1 \tag{7}$$



Fig. 3. Circuit diagram of example triode amplifier from [5].

Table 1. Distortion versus peak input voltage.

Peak Input Voltage	% 2nd HD	% THD
0.2 V	0.791	0.862
0.4 V	1.51	1.54
0.6 V	2.37	2.39
0.8 V	3.56	3.62
1.0 V	6.93	7.66



Fig. 4. Small-signal ac gain versus frequency for triode amplifier.



Fig. 5. Transient response of triode amplifier for peak sine-wave input voltages of 0.2, 0.4, 0.6, 0.8, and 1.0 V.

where a is the fraction of space current which flows in the plate. The parameter a is a function of the voltage ratio v_{PK}/v_{SK}. However, no simple form of this analytical function is known [4]. In this paper a is taken to be a constant if v_{PK} is above a threshold value. If the plate-to-cathode voltage is decreased below this threshold value, the plate current decreases rapidly to zero. In this region, the plate current is approximately independent of the voltage from the screen grid to the cathode.

The small-signal model of the pentode is given in Fig. 6(b). To simplify the figure, the interelectrode capacitors have been omitted. The transconductances g_{m1}, g_{m2}, g_{m3}, and g_{m4} and the resistances r_{pk} and r_{sk} are given by

$$g_{m1} = \frac{\partial I_P}{\partial V_{GK}} = \frac{3\mu_c I_P}{2(\mu_c V_{GK} + \mu_s V_{SK} + V_{PK})} \tag{8}$$

$$g_{m2} = \frac{\partial I_P}{\partial V_{SK}} = \frac{3\mu_s I_P}{2(\mu_c V_{GK} + \mu_s V_{SK} + V_{PK})} \tag{9}$$

$$g_{m3} = \frac{\partial I_S}{\partial V_{GK}} = \frac{3\mu_c I_S}{2(\mu_c V_{GK} + \mu_s V_{SK} + V_{PK})} \tag{10}$$

$$g_{m4} = \frac{\partial I_S}{\partial V_{PK}} = \frac{3 I_S}{2(\mu_c V_{GK} + \mu_s V_{SK} + V_{PK})} \tag{11}$$

$$r_{pk} = \left(\frac{\partial I_P}{\partial V_{PK}}\right)^{-1} = \frac{\mu_c}{g_{m1}} = \frac{\mu_s}{g_{m2}} \tag{12}$$

$$r_{sk} = \left(\frac{\partial I_S}{\partial V_{PK}}\right)^{-1} = \frac{\mu_c}{\mu_s g_{m3}} = \frac{1}{\mu_s g_{m4}}. \tag{13}$$

If the grid is biased at a voltage that is positive with respect to the cathode, the grid current cannot be neglected. In this case, a resistor must be added from the grid to the cathode to model the small-signal grid-to-cathode resistance.

Fig. 6(c) shows a *PSpice* subcircuit model for the pentode. The interelectrode capacitances are omitted to simplify the figure. As for the triode, the model is valid only for a zero or positive plate-to-cathode voltage. A series solid-state diode D_1 and resistor R_{GK} are used to approximately model the grid circuit. The SPICE code for the model is given in the following. Parameters that must be assigned numerical values are enclosed in quotations.

```
*PENTODE SUBCIRCUIT
.SUBCKT PENTODE P S G K
RGK G 1 "R_GK"
D1 1 K DM
ESP 2 0 VALUE = {V(P,K) + "μs"*V(S,K)
  + "μc"*V(G,K)}
E1 3 2 VALUE = {"K"*(PWRS(V(2),1.5)
  + PWR(V(2),1.5))/2}
E2 3 4 VALUE = {"K"*PWR("μs"*V(S,K),
  1.5)*V(P, K)/"V_A"}
E3 5 4 VALUE = {(1 - V(4,2)/ABS(V(4,2)
  + .001))/2}
R1 5 0 1K
GK S K VALUE = {V(3,2)}
GP P S VALUE = {"a"*(V(3,4)*(1 - V(5,4))
  + V(3,2)*V(5,4))}
.MODEL DM D
.ENDS
```

Fig. 7 shows the plate current and screen current versus plate voltage for a 6L6 beam power tube calculated by SPICE with the subcircuit. The curves are shown for a grid-to-cathode voltage v_{GK} that varies from 0 to -35 V in steps of -5 V. The 6L6 parameters used in the subcircuit are $K = 5.39 \times 10^{-7}$, $\mu_s = 19.3$, $\mu_c = 154$, $V_A = 50$, and $a = 0.95$. The value of V_A detemines the plate-to-cathode voltage at which the slope of the curves exhibits a sharp change.

The model parameters used for the 6L6 were derived from curves of tube characteristics given in [5]. K and μ_s were determined so that the calculated curves matched those given in [5] at the points ($V_{GK} = 0$, $V_{SK} = 400$ V, $V_{PK} = 300$ V, $I_P = 370$ mA) and ($V_{GK} = 0$ V, $V_{SK} = 50$ V, $V_{PK} = 300$ V, $I_P = 23$ mA). With the values determined for K and μ_s, μ_c was determined to make $g_{m1} = 0.0052$ A/V at the point ($V_{GK} = -18$ V,



Fig. 6. (a) Circuit symbol for pentode. (b) Small-signal model of pentode. (c) *PSpice* model for pentode.

V_{SK} = 250 V, V_{PK} = 350 V, I_P = 54 mA). Insufficient data were given in [5] to determine the parameter a. The value $a = 0.95$ was used as a good estimate for this parameter. The calculated curves show good agreement with those given in [5].

3 THE OUTPUT TRANSFORMER

An integral part of a vacuum-tube amplifier is the output transformer. All physical transformers are nonlinear devices. SPICE can be used to model nonlinear magnetic effects if the model parameters of the nonlinear core material are known [3]. Nonlinear magnetic effects are neglected here and the output transformer is represented as a set of linear coupled coils. Fig. 8 shows the circuit diagram of a transformer with primary taps for negative feedback to the screen grids. The resistance and inductance between each tap are shown as separate elements. A *PSpice* subcircuit model for the transformer is given in the following. Parameters that must be assigned numerical values are enclosed in quotations.

```
*TRANSFORMER MODEL SUBCIRCUIT
.SUBCKT TRANS P1 ST1 CT ST2 P2 S1 S2
R1 P1 1 "R1"
L1 1 SG1 "L1"
R2 SG1 2 "R2"
L2 2 CT "L2"
R3 CT 3 "R3"
L3 3 SG2 "L3"
R4 SG2 4 "R4"
L4 4 P2 "L4"
R5 S1 5 "R5"
L5 5 S2 "L5"
KALL L1 L2 L3 L4 L5 "k"
.ENDS
```

The parameter k is the mutual coupling coefficient between the coils. The polarity of the mutual coupling is determined by the order of the nodes for the inductors in the code. The circuit can be improved at high frequencies by the addition of capacitors to model capacitive coupling between the windings.

4 AN EXAMPLE SIMULATION

Fig. 9 shows the circuit diagram of an example push–pull amplifier used here to illustrate the application of the models in the design of a power amplifier. Although the topology might be typical of commercial designs, the circuit is not intended to correspond to any specific amplifier. The first stage is a differential amplifier consisting of tubes T_1 and T_2. The input signal is applied to the grid of T_1 and the feedback signal is applied to the grid of T_2. The tail resistor R_{K1} sets the



Fig. 8. Circuit model of transformer with screen taps for ultra-linear power output stage.



Fig. 7. Simulated plate and screen grid characteristics of 6L6 beam power tube.

quiescent plate current in T_1 and T_2. The second stage is a differential amplifier consisting of tubes T_3 and T_4. The tail resistor R_{K2} sets the quiescent plate current in T_3 and T_4. The plate outputs of T_3 and T_4 are coupled to the grids of T_5 and T_6 through coupling capacitors C_3 and C_4. Grid bias for T_5 and T_6 is applied through resistors R_{G5} and R_{G6}.

High-frequency compensation is provided by the input compensation networks consisting of the elements R_{C1}, C_1, R_{C2}, and C_2 in the input differential amplifier and the lead compensation capacitor C_F in the feedback network. The open-loop transfer function at low frequencies has two poles, one caused by the coupling capacitors C_3 and C_4 and one caused by the output transformer. These poles must be well separated for good low-frequency stability. To ensure this, capacitors C_3 and C_4 are made large enough in the example simulation so that the dominant low-frequency pole is determined by the transformer.

Local negative feedback is applied to the screen grids of the two pentode output tubes T_5 and T_6 from taps on the primary of the output transformer. The effect of this feedback is to reduce both the voltage gain and the output resistance of the output stage. Negative feedback from a primary tap on the output transformer to the screen grid was originally described in [6]. An amplifier design based on this principle was later called an "ultra-linear amplifier" [7]. The literature contains some interesting papers which debate the merits of this topology [8]–[11].

Fig. 10 shows a single-ended approximation to the amplifier output stage which can be used to illustrate the effect of local negative feedback to the screen grid. The pentode is represented by the small-signal model of Fig. 6(b). The circuit is simplified by assuming that the value of the cathode resistor is small enough so that it can be neglected. Let the output transformer be modeled as an ideal transformer having a turns ratio $n : 1$. Let x be the fraction of primary turns from the screen tap to ac signal ground, that is, from the screen tap to the transformer center tap, for only half of the transformer primary appears in the single-ended circuit. (The screen tap ratio for the ultralinear connection is commonly specified as a percentage impedance ratio, that is, it is specified as $x^2 \times 100\%$). If the screen current i_s is neglected, the screen voltage is given by $v_s = xv_p$. For this case it is straightforward to show that the small-signal ac voltage gain v_o/v_i and output resistance r_{out} are given by

$$\frac{v_o}{v_i} = -\frac{1}{n} \times \frac{g_{m1}\,[r_{pk} \,||\, (n^2R_L)]}{1 + xg_{m2}\,[r_{pk} \,||\, (n^2R_L)]} \tag{14}$$

$$r_{out} = \frac{1}{n^2} \times \frac{r_{pk}}{1 + xg_{m2}r_{pk}}. \tag{15}$$

For $x = 0$, the output tube is operated as a pentode; for $x = 1$, it is operated as a triode. These expressions illustrate how the small-signal voltage gain and the small-signal output resistance change between the two cases.

The SPICE code containing all numerical values for the amplifier elements is given in the Appendix. The resistors have been calculated to bias T_1 and T_2 at 0.5 mA, T_3 and T_4 at 2.5 mA, and T_5 and T_6 at 34 mA. These bias values are reported in the SPICE output file,



Fig. 10. Single-ended approximation to power amplifier output stage with pentode represented by small-signal model of Fig. 6(b).



Fig. 9. Circuit diagram of example push–pull power amplifier.

which is not included here. To minimize the complexity of the circuit so that it could be analyzed with the evaluation version of *PSpice*, the tube subcircuit models are simplified from the ones given earlier. First, the grid input resistance for positive grid-to-cathode voltage is not modeled for any of the tubes. Second, the subcircuit model for T_5 and T_6 does not model the change in slope of the output characteristics at low plate voltages. These simplifications do not affect ac frequency response calculations as long as no tube is biased in a region not modeled. Similarly, transient response calculations are valid as long as no tube enters a region not modeled. For example, the grid-to-cathode voltage must not go positive for any tube.

The transformer parameters were measured on a commercially available transformer. Although it was straightforward to measure the inductance and resistance of each winding, the coupling coefficient k proved difficult to measure because it is so close to unity. To determine k, the -3-dB upper cutoff frequency was measured with the transformer driving an 8-Ω load. A SPICE simulation of the transformer was then performed using the measured inductances and resistances. The value of k was adjusted experimentally until the SPICE simulation predicted a cutoff frequency that was equal to the measured cutoff frequency.

Fig. 11 shows the small-signal amplitude and phase responses of the amplifier versus frequency calculated with the .AC analysis. Five curves are displayed for each to show the effects of simultaneously varying C_1 through C_4 and C_F. The variation of the elements is achieved with the parametric analysis feature of *PSpice*. The element values are given by multiplying their nominal values by the parameter X, which takes on the values 0.25, 0.5, 1, 2, and 4. For the nominal values ($X = 1$), the small-signal bandwidth extends from less than 0.5 Hz on the low end to approximately 80 kHz on the high end, and the gain peaking near the upper cutoff frequency is less than 0.3 dB.

Fig. 12 shows the transient response calculated for a sine-wave input with the .TRAN analysis. The nominal values for C_1 through C_4 and C_F were used for the calculation. The waveforms for the output voltage, the grid-to-cathode voltages of T_5 and T_6, and the voltages across the cathode resistors R_{K5} and R_{K6} are shown. It can be seen from the waveforms that the grid-to-cathode voltages for T_5 and T_6 peak at approximately 0 V. If the input amplitude were increased, these voltages would go positive. In this case, the simulation would not be correct unless the SPICE code is changed to model the change in grid input resistance for a positive grid-to-cathode voltage. The voltages across R_{K5} and R_{K6} are proportional to the currents in the output tubes. The waveforms show that T_5 and T_6 cut off during one-half a cycle. Thus the output stage is biased in the class AB mode. Although the bias current in the output tubes could be increased to obtain class A operation over the entire signal swing, the maximum quiescent plate dissipation in the tubes might be exceeded.

5 CONCLUSIONS

SPICE simulations of vacuum-tube circuits can provide a powerful technique for the evaluation of circuit designs without the need to breadboard the circuits. The accuracy of the simulations depends on the accuracy of



Fig. 11. Calculated small-signal frequency responses of example power amplifier for different values of C_1 through C_4 and C_F. (a) Amplitude responses. (b) Phase responses.

the models. Device models for the triode and the pentode have been described. Several examples have been presented which illustrate the use of these models in the calculation of gain and phase versus frequency, transient response, and harmonic distortion.

6 REFERENCES

[1] A. Comarow, "A Techie's Toast: Here's Looking at You, Kit," *Smithsonian*, vol. 23, no. 12, p. 162 (1993 Mar.).

[2] R. A. Pittman, *The Tube Amp Book II*, Groove Tubes, Sylmar, CA, 1988.

[3] *PSpice Circuit Analysis*, MicroSim Corp., Irvine, CA, 1990.

[4] K. R. Spangenberg, *Fundamentals of Electron Devices*, McGraw-Hill, New York, 1957.

[5] *RCA Receiving Tube Manual*, RCA, Harrison, NJ, 1968.

[6] A. D. Blumlein, British Patent 496883 (1937 May).

[7] D. Hafler and H. I. Keroes, "An Ultra-linear Amplifier," *Audio Eng.*, pp. 15–17 (1951 Nov.).

[8] D. T. N. Williamson and P. J. Walker, "Amplifiers and Superlatives," *Wireless World*, pp. 357–361 (1952 Sept.).

[9] F. Langford-Smith and A. R. Chesterman, "Ultra Linear Amplifiers" (3 parts), *Radiotronics*, vol. 20, nos. 5, 6, 7 (1955 May, June, July).

[10] "Tetrodes with Screen Feedback," *Wireless World*, pp. 24–26 (1956 Jan.).

[11] D. M. Leakey and R. B. Gilson, "UL Output Transformers," *Wireless World*, pp. 29–32 (1956 Jan.).

APPENDIX

The *PSpice* code for the example amplifier simulation is given here. Disabled lines are preceded by an asterisk.

```
EXAMPLE AMPLIFIER SIMULATION
VIN 1 0 AC 1V
*VIN 1 0 SIN (0,1,1K,0,0,0)
VPP1 21 0 DC 200
VPP2 22 0 DC 300
VPP3 25 0 DC 400
VKK 23 0 DC -300
VGG 24 0 DC -39
R1 1 0 20K
R2 1 2 1K
RP1 3 21 100K
RK1 4 23 300K
RP2 6 21 100K
RF1 5 0 1K
RP3 9 22 39K
RC1 7 21 3.9K
RK2 10 0 30K
RP4 11 22 39K
RC2 8 21 3.9K
RG5 12 24 390K
RG6 13 24 390K
RF2 20 5 20K
RK5 14 0 10
RK6 15 0 10
RL 20 0 8
.PARAM X = 1
C1 3 7 {0.001U*X}
C2 6 8 {0.001U*X}
```



(a)



(b)

Fig. 12. Calculated transient response for example power amplifier for sine-wave input. (a) $V(20)$—output voltage; $V(13, 15)$—grid-to-cathode voltage of T_6; $V(12,14)$—grid-to-cathode voltage of T_5. (b) $V(15)$—cathode voltage of T_6; $V(14)$—cathode voltage of T_5.

```
C3 9 12 {2.2U*X}
C4 11 13 {2.2U*X}
CF 5 20 {22P*X}
.STEP PARAM OCT X 0.25 4 1
XT1 3 2 4 12AX7
XT2 6 5 4 12AX7
XT3 9 3 10 12AT7
XT4 11 6 10 12AT7
XT5 16 17 12 14 6L6
XT6 19 18 13 15 6L6
XT7 19 18 25 17 16 20 0 TRANS
.SUBCKT 12AX7 P G K
E1 2 0 VALUE = {V(P,K) + 83.5*V(G,K)}
R1 2 0 1K
GP P K VALUE = {1.73E-6*(PWR)V(2),1.5)
  + PWRS(V(2),1.5))/2}
CGP G P 1.7P
CGK G K 1.6P
GPK P K 0.4P
.ENDS
.SUBCKT 12AT7 P G K
E1 2 0 VALUE = {V(P,K) + 44.8*V(G,K)}
R1 2 0 1K
GP P K VALUE = {8.31E-6*(PWR(V(2),1.5)
  + PWRS(V(2),1.5))/2}
CGP G P 1.5P
CGK G K 2.2P
CPK P K 0.45P
.ENDS
.SUBCKT 6L6 P S G K
ESP 2 0 VALUE = {V(P,K) + 19.3*V(S,K)
  + 154*V(G,K)}
E1 3 2 VALUE = {5.39E-7*(PWR(V(2),1.5)
  + PWRS(V(2),1.5))/2}
R1 3 0 1K
GK S K VALUE = {V(3,2)}
GP P S VALUE = {0.95*V(3,2)}
CGP G P 0.6P
CGK G K 5P
CGS G S 5P
CPK P K 6.5P
.ENDS
.SUBCKT TRANS P1 ST1 CT ST2 P2 S1 S2
R1 P1 1 29.7
L1 1 ST1 4.63
R2 ST1 2 27.7
L2 2 CT 2.9
R3 CT 3 26.9
L3 3 ST2 2.9
R4 ST2 4 37.9
L4 4 P2 4.62
R5 S1 5 0.487
KALL L1 L2 L3 L4 L5 0.9988
.ENDS
.AC DEC 20 0.1 1MEG
*.TRAN (0.01M,2M,0,0,0.01M)
.PROBE
.END
```

THE AUTHOR



W. Marshall Leach, Jr., received B.S. and M.S. degrees in electrical engineering from the University of South Carolina, Columbia, in 1962 and 1964, and a Ph.D. degree in electrical engineering from the Georgia Institute of Technology in 1972.

In 1964 he worked at the National Aeronautics and Space Administration in Hampton, VA. From 1965 to 1968 he served as an officer in the U.S. Air Force. Since 1972 he has been a faculty member at the Georgia Institute of Technology, where he is presently professor of electrical engineering. Dr. Leach teaches courses in applied electromagnetics and electronic design. He is a fellow of the Audio Engineering Society and a senior member of the IEEE.